

**Math 2250 Maple Project 5: Linear Algebra  
F2010**

**Due date:** See the internet due dates. Maple lab 5 has four problems: L5.1, L5.2, L5.3, L5.4. Examples of the maple coding required appears in four examples at the end of this document.

**References:** Code in maple appears in 2250mapleL5-F2010.txt at URL <http://www.math.utah.edu/~gustafso/>. This document: 2250mapleL5-F2010.pdf.

**Problem L5.1. (Matrix Algebra)**

Define  $A = \begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{pmatrix}$ ,  $B = \begin{pmatrix} 2 & 1 & 0 \\ 1 & 2 & 1 \\ 0 & 1 & 2 \end{pmatrix}$ ,  $\mathbf{v} = \begin{pmatrix} 1 \\ 2 \\ 3 \end{pmatrix}$  and  $\mathbf{w} = \begin{pmatrix} -1 \\ 4 \\ 1 \end{pmatrix}$ . Create a worksheet in maple which states this problem in text, then defines the four objects. The worksheet should contain text, maple code and displays. Continue with this worksheet to answer (1)–(7) below. Submit problem L5.1 as a worksheet printed on 8.5 by 11 inch paper. See Example 1 for maple commands.

- (1) Compute  $AB$  and  $BA$ . Are they the same?
- (2) Compute  $A + B$  and  $B + A$ . Are they the same?
- (3) Let  $C = A + B$ . Compare  $C^2$  to  $A^2 + 2AB + B^2$ . Explain why they are different.
- (4) Compute transposes  $C_1 = (AB)^T$ ,  $C_2 = A^T$  and  $C_3 = B^T$ . Find a matrix equation for  $C_1$  in terms of  $C_2$  and  $C_3$ . Verify the equation.
- (5) Solve for  $\mathbf{X}$  in  $B\mathbf{X} = \mathbf{v}$  by maple commands `rref`, `linsolve`, `inverse`.
- (6) Solve  $A\mathbf{Y} = \mathbf{v}$  for  $\mathbf{Y}$ . Do an answer check using `linsolve`.
- (7) Solve  $A\mathbf{Z} = \mathbf{w}$ . Explain your answer using the three possibilities for a linear system. Discuss the possible maple reports for (1) no solution case, (2) unique solution, (3) infinitely many solutions.

**Problem L5.2. (Independent Columns)**

$$\text{Let } A = \begin{pmatrix} 1 & 1 & 1 & 2 & 6 \\ 2 & 3 & -2 & 1 & -3 \\ 0 & 1 & -4 & -3 & -15 \\ 1 & 2 & -3 & -1 & -9 \end{pmatrix}.$$

Find independent vectors which have the same span as the columns of  $A$  using the following methods.

**Method 1.** Find the pivot columns of  $A$ . See Example 2.

**Method 2.** The maple command `colspace(A)`.

The first method is equivalent to finding a largest set of independent vectors from the list of 5 vectors formed from the columns of  $A$ . The answer is a basis of 2 vectors. The span of these 2 vectors equals the span of the 5 column vectors of  $A$ . The second method finds another basis of 2 vectors, which is generally different, but equivalent in the sense described in the next part.

**Problem L5.3. (Equivalent Bases)**

Let

$$\mathbf{v}_1 = \begin{pmatrix} 1 \\ 2 \\ 0 \\ 1 \end{pmatrix}, \quad \mathbf{v}_2 = \begin{pmatrix} 1 \\ 3 \\ 1 \\ 2 \end{pmatrix}, \quad \mathbf{w}_1 = \begin{pmatrix} 1 \\ 0 \\ -2 \\ -1 \end{pmatrix}, \quad \mathbf{w}_2 = \begin{pmatrix} 0 \\ 1 \\ 1 \\ 1 \end{pmatrix}.$$

Verify that the two bases  $\mathcal{B}_1 = \{\mathbf{v}_1, \mathbf{v}_2\}$  and  $\mathcal{B}_2 = \{\mathbf{w}_1, \mathbf{w}_2\}$  are **equivalent**. This means that each vector in  $\mathcal{B}_1$  is a linear combination of the vectors in  $\mathcal{B}_2$ , and conversely, that each vector in  $\mathcal{B}_2$  is a linear combination of the vectors in  $\mathcal{B}_1$ . Briefly,

$$\text{span}\{\mathbf{v}_1, \mathbf{v}_2\} = \text{span}\{\mathbf{w}_1, \mathbf{w}_2\}.$$

---

Staple this page on top of the maple work sheets.

### Problem L5.4. (Matrix Equations)

Let  $A = \begin{pmatrix} 8 & 10 & 3 \\ -3 & -5 & -3 \\ -4 & -4 & 1 \end{pmatrix}$ ,  $T = \begin{pmatrix} 1 & 0 & 0 \\ 0 & -2 & 0 \\ 0 & 0 & 5 \end{pmatrix}$ . Let  $P$  denote a  $3 \times 3$  matrix. Define  $\lambda_1 = 1$ ,  $\lambda_2 = -2$  and  $\lambda_3 = 5$ .

Assume the following result:

**Lemma 1.** The equality  $AP = PT$  holds if and only if the columns  $\mathbf{v}_1$ ,  $\mathbf{v}_2$ ,  $\mathbf{v}_3$  of  $P$  satisfy  $A\mathbf{v}_1 = \lambda_1\mathbf{v}_1$ ,  $A\mathbf{v}_2 = \lambda_2\mathbf{v}_2$ ,  $A\mathbf{v}_3 = \lambda_3\mathbf{v}_3$ . [proved after Example 4]

(a) Determine three specific columns for  $P$  such that  $\det(P) \neq 0$  and  $AP = PT$ . These columns contain only numbers – no symbols allowed! Infinitely many answers are possible. See Example 4 for the maple method that determines a column of  $P$ .

(b) After reporting the three columns, check the answer by computing  $AP - PT$  (it should be zero) and  $\det(P)$  (it should be nonzero).

**Example 1.** Let  $A = \begin{pmatrix} 1 & 2 & 3 \\ 2 & -1 & 1 \\ 3 & 0 & -1 \end{pmatrix}$  and  $\mathbf{b} = \begin{pmatrix} 9 \\ 8 \\ 3 \end{pmatrix}$ . Create a `maple` work sheet. Define and display matrix  $A$  and vector  $\mathbf{b}$ . Then compute

- (1) The inverse of  $A$ .
- (2) The augmented matrix  $C = \mathbf{aug}(A, \mathbf{b})$ .
- (3) The reduced row echelon form  $R = \mathbf{rref}(C)$ .
- (4) The column  $\mathbf{X}$  of  $R$  which solves  $A\mathbf{X} = \mathbf{b}$ .
- (5) The matrix  $A^3$ .
- (6) The transpose of  $A$ .
- (7) The matrix  $A - 3A^2$ .
- (8) The solution  $\mathbf{X}$  of  $A\mathbf{X} = \mathbf{b}$  by two methods different than (4).
- (9) Find a matrix  $F$  such that  $F\mathbf{x} = \mathbf{b}$  has no solution. Explain why `linsolve` prints nothing.
- (10) Compute  $A^T A$ ,  $(A^T A)^{-1}$ ,  $A^{-1}(A^{-1})^T$ .

**Solution:** A lab instructor or classmate can help you to create a blank work sheet in `maple`, enter code and print the work sheet. The code to be entered appears below. To get help, enter `?linalg` into a worksheet, then select commands that match ones below.

```
with(linalg):
A:=matrix([[1,2,3],[2,-1,1],[3,0,-1]]);
b:=vector([9,8,3]);
print("(1)"); inverse(A);
print("(2)"); C:=augment(A,b);
print("(3)"); R:=rref(C);
print("(4)"); X:=col(R,4);
print("(5)"); evalm(A^3);
print("(6)"); transpose(A);
print("(7)"); evalm(A-3*(A^2));
print("(8)"); X:=linsolve(A,b); X:=evalm(inverse(A) &* b);
print("(9)"); F:=matrix([[1,2,3],[2,-1,1],[0,0,0]]);linsolve(F,b);
# Nothing is printed, because of a signal equation "0=3".
print("(10)"); evalm(transpose(A) &* A); evalm(inverse(transpose(A) &* A));
evalm(inverse(A)&*transpose(inverse(A)));
```

**Example 2.** Let  $A = \begin{pmatrix} 1 & 1 & 1 & 2 & 6 \\ 2 & 3 & -2 & 1 & -3 \\ 3 & 5 & -5 & 1 & -8 \\ 4 & 3 & 8 & 2 & 3 \end{pmatrix}$ .

- (1) Find a basis for the column space of  $A$ . This means: find a largest list of independent columns of  $A$ .
- (2) Find a basis for the row space of  $A$ .
- (3) Find a basis for the nullspace of  $A$ . This is the list of vector partials  $\partial_{t_1}\mathbf{x}$ ,  $\partial_{t_2}\mathbf{x}$ ,  $\dots$  applied to the general solution  $\mathbf{x}$  of  $A\mathbf{x} = \mathbf{0}$ , which is obtained from the *last frame algorithm*.
- (4) Find  $\mathbf{rank}(A)$  and  $\mathbf{nullity}(A)$ . They are the number of lead variables and the number of free variables for the problem  $A\mathbf{x} = \mathbf{0}$ , respectively.
- (5) Find the dimensions of the nullspace, row space and column space of  $A$ .

**Solution:** The theory applied: *The columns of  $B$  corresponding to the leading ones in  $\mathbf{rref}(B)$  are independent and form a basis for the column space of  $B$ .* These columns are called the **pivot columns** of  $B$ . The meaning is

$$\mathbf{span}\{\text{all columns of } B\} = \mathbf{span}\{\text{pivot columns of } B\}.$$

A list of vectors is called a **basis** provided it is **independent** and **spans**.

Results for the row space of  $A$  are obtained by replacing  $B$  by the transpose of  $A$ . In particular, the row space of  $A$  is spanned by the pivot columns of  $B = A^T$ .

The `maple` code which applies is

```
with(linalg):
A:=matrix([[ 1, 1, 1, 2, 6],
           [ 2, 3,-2, 1,-3],
           [ 3, 5,-5, 1,-8],
           [ 4, 3, 8, 2, 3]]);
print("(1)"); C:=rref(A); # leading ones in columns 1,2,4
           BASIScolumnspace=col(A,1),col(A,2),col(A,4);
print("(2)"); F:=rref(transpose(A)); # leading ones in columns 1,2,3
           BASISrowspace=row(A,1),row(A,2),row(A,3);
print("(3)"); nullspace(A); linsolve(A,vector([0,0,0,0]));
print("(4)"); RANK=rank(A); NULLITY=coldim(A)-rank(A);
print("(5)"); DIMnullspace=coldim(A)-rank(A); DIMrowspace=rank(A);
           DIMcolumnspace=rank(A);
```

**Example 3.** Let  $A = \begin{pmatrix} 1 & 1 & 1 & 2 & 6 \\ 2 & 3 & -2 & 1 & -3 \\ 3 & 5 & -5 & 1 & -8 \\ 4 & 3 & 8 & 2 & 3 \end{pmatrix}$ . Verify that the following column space bases of  $A$  are equivalent.

$$\mathbf{v}_1 = \begin{pmatrix} 1 \\ 2 \\ 3 \\ 4 \end{pmatrix}, \quad \mathbf{v}_2 = \begin{pmatrix} 1 \\ 3 \\ 5 \\ 3 \end{pmatrix}, \quad \mathbf{v}_3 = \begin{pmatrix} 2 \\ 1 \\ 1 \\ 2 \end{pmatrix},$$

$$\mathbf{w}_1 = \begin{pmatrix} 1 \\ 0 \\ 0 \\ -3 \end{pmatrix}, \quad \mathbf{w}_2 = \begin{pmatrix} 0 \\ 1 \\ 0 \\ 17 \end{pmatrix}, \quad \mathbf{w}_3 = \begin{pmatrix} 0 \\ 0 \\ 1 \\ -9 \end{pmatrix}.$$

**Solution:** We will use this result:

**Lemma 2.** Bases  $\{\mathbf{v}_1, \mathbf{v}_2, \mathbf{v}_3\}$  and  $\{\mathbf{w}_1, \mathbf{w}_2, \mathbf{w}_3\}$  are equivalent bases if and only if the augmented matrices  $F = \mathbf{aug}(\mathbf{v}_1, \mathbf{v}_2, \mathbf{v}_3)$ ,  $G = \mathbf{aug}(\mathbf{w}_1, \mathbf{w}_2, \mathbf{w}_3)$  and  $H = \mathbf{aug}(F, G)$  satisfy the rank condition  $\mathbf{rank}(F) = \mathbf{rank}(G) = \mathbf{rank}(H) = 3$ .

The proof appears below.

The `maple` code which applies is

```
with(linalg):
A:=matrix([[ 1, 1, 1, 2, 6],
           [ 2, 3,-2, 1,-3],
           [ 3, 5,-5, 1,-8],
           [ 4, 3, 8, 2, 3]]);
v1:=vector([1,2,3,4]); v2:=vector([1,3,5,3]); v3:=vector([2,1,1,2]);
w1:=vector([1, 0, 0, -3]); w2:=vector([0, 1, 0, 17]); w3:=vector([0, 0, 1, -9]);
F:=augment(v1,v2,v3);
G:=augment(w1,w2,w3);
H:=augment(v1,v2,v3,w1,w2,w3);
rank(F); rank(G); rank(H);
```

We remark that the two bases in the example were discovered from the `maple` code

```

rref(A); # pivot cols 1,2,4
v1:=col(A,1); v2:=col(A,2); v3:=col(A,4);
B:=rref(transpose(A)); # pivot cols 1,2,3
w1:=row(B,1); w2:=row(B,2); w3:=row(B,3);

```

**Proof of Lemma 2.**

**Proof:** Let's justify part of the test, showing only half the proof:  $\mathbf{rank}(F) = \mathbf{rank}(G) = \mathbf{rank}(H) = n$  implies the bases are equivalent.

The equation  $\mathbf{rref}(F) = EF$  holds for  $E$  a product of elementary matrices. Then  $\mathbf{rref}(H) = EH$  has to have  $n$  leading ones, because of  $F$  in the first  $n$  columns, and the remaining rows of  $\mathbf{rref}(H)$  are zero, because  $\mathbf{rank}(H) = n$ . Therefore, the first  $n$  columns of  $H = \mathbf{aug}(F, G)$  are the pivot columns of  $H$ . The non-pivots of  $H$  are just the columns of  $G$ , and by the pivot theory, they are linear combinations of the pivot columns, namely, the columns of  $F$ . We can multiply  $H$  by a permutation matrix  $P$  which effectively swaps  $F$  and  $G$ . Already,  $HP$  has the  $n$  independent columns of  $F$ , so its rank is at least  $n$ . But its other columns are linear combinations of these columns, so the rank is exactly  $n$ . Now we argue by symmetry that the columns of  $F$  are linear combinations of the columns of  $G$ , using  $HP$  instead of  $H$ .

The first half of the proof is complete. The other half is left to the reader.

**Example 4.** Let  $A = \begin{pmatrix} 1 & 2 & 3 \\ 2 & -1 & 1 \\ 3 & 0 & 0 \end{pmatrix}$ . Solve the equation  $A\mathbf{x} = -3\mathbf{x}$  for  $\mathbf{x}$ .

**Solution.** Let  $\lambda = -3$ . The idea is to write the equation  $A\mathbf{x} = \lambda\mathbf{x}$  as a homogeneous problem  $(A - \lambda I)\mathbf{x} = \mathbf{0}$ .

The trick is to move  $\lambda\mathbf{x}$  from the RHS to the LHS of the equation, then re-write  $\lambda\mathbf{x}$  as  $\lambda I\mathbf{x}$ , where  $I$  is the identity matrix. Then  $\mathbf{x}$  is a common factor, and the matrix equation can be written as  $(A - \lambda I)\mathbf{x} = \mathbf{0}$ .

Define  $B = A - \lambda I$ . The homogeneous equation  $B\mathbf{x} = \mathbf{0}$  always has the solution  $\mathbf{x} = \mathbf{0}$ . It has a nonzero solution  $\mathbf{x}$  if and only if there are infinitely many solutions, in which case the solutions are found by a frame sequence to  $\mathbf{rref}(B)$ . The `maple` details appear below. The basis vectors for  $B\mathbf{x} = \mathbf{0}$  are obtained in the usual way, by taking partial derivatives on the general solution with respect to the symbols  $t_1, t_2, \dots$ . In this case, there is just one basis vector

$$\begin{pmatrix} -2 \\ 1 \\ 2 \end{pmatrix}.$$

```

with(linalg):
A:=matrix([[1,2,3],[2,-1,1],[3,0,0]]);
B:=evalm(A-(-3)*diag(1,1,1));
linsolve(B,vector([0,0,0]));
# ans: t_1*vector([-2,1,2])
# Basis == partial on t_1 == vector([-2,1,2])

```

**Proof of Lemma 1.** Define  $r_1 = 1, r_2 = -2, r_3 = 5$ . Assume  $AP = PT, P = \mathbf{aug}(\mathbf{v}_1, \mathbf{v}_2, \mathbf{v}_3)$  and  $T = \mathbf{diag}(r_1, r_2, r_3)$ . The definition of matrix multiplication implies that  $AP = \mathbf{aug}(A\mathbf{v}_1, A\mathbf{v}_2, A\mathbf{v}_3)$  and  $PT = \mathbf{aug}(r_1\mathbf{v}_1, r_2\mathbf{v}_2, r_3\mathbf{v}_3)$ . Then  $AP = PT$  holds if and only if the columns of the two matrices match, which is equivalent to the three equations  $A\mathbf{v}_1 = r_1\mathbf{v}_1, A\mathbf{v}_2 = r_2\mathbf{v}_2, A\mathbf{v}_3 = r_3\mathbf{v}_3$ . The proof is complete.

**End of Maple Lab 5 Linear Algebra.**